
1 はじめに
 ストリーミング SIMD拡張命令 2(SSE2、Streaming SIMD Extensions 2)では、SIMD(Single
Instruction Multiple Data)倍精度浮動小数点命令、および SIMD整数命令が IA-32 インテル® アー

キテクチャに新しく導入された。倍精度 SIMD命令による機能拡張の方法は、ストリーミング
SIMD拡張命令(SSE)で導入された単精度 SIMD命令による機能拡張とよく似ている。128ビッ
トの整数 SIMD拡張命令は、64ビットの整数 SIMD拡張命令の完全なスーパーセットで、より
多くの整数データ型、整数と浮動小数点間のデータ型変換、キャッシュとシステム・メモリの

効果的使用をサポートする命令が追加されている。これらの命令は、3Dグラフィックス、リ
アルタイムの物理的な現象、空間的(3D)オーディオ、ビデオ・エンコーディング/デコーディン
グ、暗号化、および科学計算アプリケーションによく使用される演算を高速化する。このアプ

リケーション・ノートでは、128ビット SIMD整数命令を使用して、被乗数のサイズが非常に
大きい乗算を実行する方法について説明する。また、SSE2命令を利用したコードの例も示す。

 大数の乗算とは、本書の用語では、それぞれの被乗数のサイズが目的のマシンの命令セットの
ネイティブ・サイズより大きい場合の 2つの数値の乗算である。基本的なアルゴリズムは、小
学校で教わる、位取り記数法を使用して 2つの数を乗算する方法と同じである。ただし、この
アルゴリズムは、非常に大きな整数値を扱えるように拡張されている。本書では、SSE2を使
用してこのアルゴリズムのパフォーマンスを大きく向上させるコーディング手法について説明

する。

 極端に大きな数を扱える、さらに高度なアルゴリズムについては、Knuth, 98[2]を参照のこと。
これらのアルゴリズムの中には、本書で説明する手法を使用できる、小さいサイズの乗算に基

づくものがある。それ以外のアルゴリズムは、オーバーヘッドが大きすぎるため、入力値のサ

イズが極端に大きい場合を除いて、SSE2によって最適化された位取り記数法アルゴリズムよ
り処理が遅くなる。

2 大数の乗算
2つの大数を乗算するためのアルゴリズムは、算数が得意な小学校の生徒でもよく知っている。
ここで、「大数」とは、数値が大きすぎるため、乗算を実行すると、目的のマシンの命令セッ

ト上でオーバーフローが発生するという意味である。

図 1に、一般的な「乗算 - 加算 - キャリー」アルゴリズムの構成を示す。また、第 5章の例 1
に、最適化されていない Cコードを示す。図 1では、添字の番号は従来の順序(最上位バイト
が左側)に従っているが、実際のコードでは、リトル・エンディアン・マシンの自然な順序に
合わせてバイト・オーダが逆転される。例の Cコードは、逆転された順序を示している。ただ
し、このコードは、手計算とは異なり、部分和を明示的に記録しない。アルゴリズムの進行に

つれて、部分和が中間合計 Tに加算されていき、最後に Tで結果全体が得られる。

本書で説明するコードは、このアルゴリズムを修正したものである。このコードは、SSE2に
よるパフォーマンスの向上を活用できるように、多少修正されている。具体的な変更点につい

ては、「2.3 大数の乗算のコーディング」で説明する。

図1:  乗算アルゴリズムの説明図

2.1 大数の乗算の応用分野
この関数は、整数論暗号アプリケーションによく使用されており、特に公開鍵暗号に頻繁に使

用される。インターネットの成長とともに、公開鍵暗号システムの重要性は急速に高まった。

公開鍵暗号は、安全な情報交換のための合意に基づくチャネルを確立できない当事者の間に、

安全な通信の基礎を提供する。大数の乗算に依存するアルゴリズムとプロトコルの例には、次

のものがある。

• RSA公開鍵暗号およびシグニチャ

• Diffie-Hellman鍵交換

• ElGamal公開鍵暗号

• DSA公開鍵シグニチャ

• 素ガロア域上でのモジュラ算術に基づく楕円曲線公開鍵システム

• SSL(Secure Sockets Layer)。これは、既に触れた暗号システムを間接的に応用した技術であ
り、インターネット上の電子商取引に広く使用されている。

これらの暗号システムについては、B. Schneier, 96[1]を参照のこと。

大数の乗算は、極度に高い精度を必要とする科学計算および研究アプリケーションでも重要で

ある。

7 6 5 4 3 2 1 0
被乗数 A  

B  

0,7 0,6 0,5 0,4 0,3 0,2 0,1 0,0
 1,7 1,6 1,5 1,4 1,3 1,2 1,1 1,0
2,7 2,6 2,5 2,4 2,3 2,2 2,1 2,0

部分積 3,7 3,6 3,5 3,4 3,3 3,2 3,1 3,0
4,7 4,6 4,5 4,4 4,3 4,2 4,1 4,0

5,7 5,6 5,5 5,4 5,3 5,2 5,1 5,0
6,7 6,6 6,5 6,4 6,3 6,2 6,1 6,0

7,7 7,6 7,5 7,4 7,3 7,2 7,1 7,0

T 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1 0  

添字

合計

2.2 大数の乗算の背景情報
この関数は古代から使用されており、考案者は不明である。

2.3 大数の乗算のコーディング
SSE2は、このアルゴリズムのパフォーマンスを向上させる、いくつかの新しい命令を用意し
ている。これらの命令によって、乗算全体の実行に必要な計算の回数を大幅に減らすことがで

きる。これらの命令についての詳細は、『インテル® 
アーキテクチャ・ソフトウェア・デベ

ロッパーズ・マニュアル、中巻: 命令セット・リファレンス』を参照のこと。

• pmuludq – Packed Multiply Double-Word to Quad-Word Unsigned。この命令は、デスティネー
ション・レジスタ内の 2つの符号なしダブルワードに、ソース・レジスタ内の 2つの符号
なしダブルワードを掛ける。2つの乗算の 64ビットの結果が、デスティネーション・レジ
スタの上位半分と下位半分に書き込まれる。32ビット乗算を使用することで、16ビット乗
算に対して 1/4の命令数で同じ乗算を実行できる。

• paddq – Packed Add Quad-word 64 bits。この命令は、ソース・レジスタとデスティネーショ
ン・レジスタの下位半分と上位半分を加算して、結果をデスティネーション・レジスタに

格納する。

次の命令を使用して、乗算を実行する際に上記の SSE2のメリットを活用できる。

• pshufd – Packed Shuffle Double-Word。この命令は、ソース・レジスタ内のワードを自由に
並べ替えたり組み合わせたりして、デスティネーション・レジスタに格納する。

• pslldq, psrldq – Packed Shift Left/Right Logical Double-Quadword。この命令は、128ビッ
ト・レジスタ全体を、指定したバイト数だけ左または右に論理シフトする。

• psllq, psrlq – Packed Shift Left/Right Logical。この命令は、128ビット・レジスタの上位
半分または下位半分を、左または右に論理シフトする。シフトするビット数は、命令内で

指定される。

• movdqua – Move Double QuadWord Aligned。この命令は、16バイト・データをロードまた
はストアする。このデータは 16バイト境界にアライメントされている必要がある。

2.3.1 実現技法

このアルゴリズムの最適化されたコードには、以下の最適化手法が使用されている。

• pmuludqを使用して乗算の回数を減らし、paddqを使用して加算を実行する。

この手法は、大数の乗算に SSE2の使用によるパフォーマンスの向上の主要な要因である。
これらの命令を使用せずに 16ビット乗算を実行した場合、1回の 32ビット×32ビット乗
算で行われる処理を実現するには、pmuludqによって 1回の操作を実行する代わりに、4
回の乗算と 4回の加算を実行しなければならない。

• 可能な限り SSE2の 128ビット命令を使用する。

この手法の明らかなメリットは、1回の命令分のコストで、複数の計算を実行できること
である。デメリットは、計算から「自然な」SIMD構造が失われることである。したがっ

て、SIMDを活用するためのデータ構造の修正に、若干のオーバーヘッドが生じる。SSE2
には多くのデータ操作命令があるため、この作業の負担は軽減される。

• 大数の乗算を小さな部分計算に分割する。

このアルゴリズムは、それぞれの被乗数から一度に 4つのダブルワードを読み出し、2つ
の 16バイト・レジスタに格納する。これによって、データの各部分をロードする回数が減
少する。図 2は、このアルゴリズムの内側ループの各部分計算で計算されるデータの領域
を示している。

図2:  部分計算への分割

• すべての部分積を計算してから、キャリーの伝搬を実行する。

この手法によって、このアルゴリズムの最適化されていないコードの中間手順に必要な、

ジャッグル操作の数を減らせる。手順は次のとおりである。各部分積を計算する際に、部

分積を下位半分(和)と上位半分(キャリー)に分割する。各列に、一時的な 64ビットのキャ
リー/和のペアが保持される。すべての部分和がこの方法で累積され、合計された後、最後
に値が渡されて、すべての列にキャリーを伝搬し、最終結果が得られる。

部分積をキャリー部分と和部分に分割し、それらを別々に合計することで、部分積の各行

にキャリーを伝搬するオーバーヘッドを避けられ、オーバーヘッドの問題を回避できる。

この方法でパフォーマンスが向上するのは、paddq命令によって 64ビット・データを直接
に加算できるからである。この手法は、実際に検出される可能性のあるあらゆるサイズの

被乗数に対して有効である。

ただし、場合によっては、もう 1つの手法を使用して、適切な条件下でのパフォーマンス
をさらに向上させることができる。被乗数の長さが前もってわかっている場合は、コード

の適切な修正によって各桁に置かれるビット数を減らすと、キャリー/和のペアへの分割が
不要になる。これにより、複数の部分積をオーバーフローなしで加算できる。例として、

1024×1024ビットの乗算を考える。それぞれの 1024ビット・データを 29ビット部分に分
割する。この場合、数値を表現するのに 36桁が必要になるが、部分積の合計は 64ビット
に納まるため、オーバーフローが発生する可能性はない。中間結果を上手に調整すれば、2
つの乗算を実行した後に paddqを使用して直接に加算を実行でき、キャリーと和のペアへ

7 6 5 4 3 2 1 0
被乗数 A  

B  

0,7 0,6 0,5 0,4 0,3 0,2 0,1 0,0
 1,7 1,6 1,5 1,4 1,3 1,2 1,1 1,0
2,7 2,6 2,5 2,4 2,3 2,2 2,1 2,0

部分積 3,7 3,6 3,5 3,4 3,3 3,2 3,1 3,0
4,7 4,6 4,5 4,4 4,3 4,2 4,1 4,0

5,7 5,6 5,5 5,4 5,3 5,2 5,1 5,0
6,7 6,6 6,5 6,4 6,3 6,2 6,1 6,0

7,7 7,6 7,5 7,4 7,3 7,2 7,1 7,0

T 15 14 13 12 11 10 9 8 7 6 5 4 3 2 1 0  

添字

合計

の分割は不要になる。これによって、必要な乗算の総数は(この場合は 1024から 1296に)増
えるが、分割操作のコストはかなり大きいため、乗算の増加のコストよりパフォーマンス

の向上の方が大きくなる。

3 パフォーマンス

3.1 パフォーマンスの向上/改善
SSE2の使用によって、大数の乗算のパフォーマンスは大きく向上する。パフォーマンスの向
上の主要な要因は、次の 2つである。

• 32ビット乗算命令と 64ビット加算命令によって、1回の計算で問題の大きなブロックを
解くことができる。

• SIMD命令の使用によって、並列性が向上する。

3.2 考察
新しいアーキテクチャ上の拡張機能によるパフォーマンスの向上を最大限に活用するには、最

適化されていないコードに対して、アルゴリズムをかなり修正する必要がある。

4 結論
このように、SSE2を使用して大数の乗算アルゴリズムを実行すれば、大きなメリットが得ら
れる。この関数は公開鍵暗号に頻繁に使用されるため、SSE2は、公開鍵暗号アプリケーショ
ンのパフォーマンスを大きく向上させるはずである。

5 C/C++のコード例
次のコードは、このアルゴリズムの最適化されていないコードを示している。このコードは、

本書の他の章で詳しく説明した最適化手法を使用していない。

例1:  最適化されていないアルゴリズム

// A[0..i-1] = Multiplicand 1

// B[0..j-1] = Multiplicand 2

// T[0..i+j-1] = Result

// WORDS = number of words in multiplicand

// Assumptions: T is initialized to zero

//              A, B are in LSB to MSB order

unsigned short A[WORDS];

unsigned short B[WORDS];

unsigned short T[WORDS];

BigMultiply ()

{

unsigned short i,j,carry;

unsigned long result;

for(i = 0; i < WORDS; i++)

{

for(carry = 0, j = 0; j < WORDS, j++)

{

result = A[i] * B[j] + t[i+j] + carry;

t[i+j] = LoWord(result);

carry =  HiWord(result);

}

}

T[i+j] = carry;

}

6 SSE2アセンブリ・コードの例
このコードは、本書で説明した最適化手法を使用している。このコードでは、被乗数と結果が

16バイトにアライメントされ、それぞれの被乗数のサイズが 128ビットの倍数であると仮定す
る。この関数は、異なるサイズの被乗数を扱えるように簡単に拡張できる。被乗数の長さが

128ビットでない場合は、変則的な条件を処理するためのコードを追加する必要がある。

32ビット形式から 29ビット形式への変換(およびその逆の変換)は、ここには示していない。
多くのアプリケーションでは、この最適化のオーバーヘッドは、多数の反復の間に軽減される。

このコードは、部分積を保持するために 2つの配列を使用する。これは、128ビット・レジス
タに対する、16バイトにアライメントが合わないロードとストアを避けるために行われる。こ
れを行わない場合は、大きなパフォーマンスのペナルティが発生する。

アルゴリズムの簡単な要約(多くの細かい部分を省略した疑似コード)を次に示す。

例2:  大数の乗算の疑似コード

// Some variables and functions shown below do not exist per se in

// the actual code, but are inserted here for clarity

For each set of four values in A

For each set of four values in B

For each possible pair A[i],B[j] (taken two at time)

Compute A[i]*B[j]

Split product into 32-bit carry/sum portions

Add product to the appropriately aligned set of partial products

Carry = 0

For each column of output

Add the corresponding two columns of partial products

Put the lower 29 bits into the output column

Carry = Upper 35 bits of the results

Output the carry

例3:  大数の乗算のアセンブリ・コード

#define DIGITS 36

#define DIGITSTIMESTWO 72

#define DIGITSMINUSFOUR 32

// This version uses delayed carry and 29-bit multiplies.

// Specifically, no carries are done until all of the partial sums have 

// completed.  By using 29 bits only, we guarantee that the sum of all the

// partial sums will fit in 64 bits.  This makes it much cleaner to support

// the sequence of

// PSHUFD (align two multiplicand digits in halves of two XMM registers)

// PMULUDQ (do the multiply)

// PADDQ  (store the result in the correct partial sum)

// Also, we use two result arrays that we can always do 16-byte aligned

// stores and load.  At the end, the two result arrays are 'stitched'

// together at the same time the carry is being propagated.

// The added number of multiplications required for the shortening from

// 32 to 29 bits is easily absorbed by the speedup in the rest of the 

// algorithm.

// 

int multiply ( int *a, int *b, int *t, int *u) 

{ 

_asm {

// t and u are intermediate results.

// t will be used for the final results at the end

mov eax, dword ptr [a]

mov ebx, dword ptr [b]

mov ecx, dword ptr [t]

mov edx, dword ptr [u]

mov eax,0

outer:

push ebx

mov ebx, dword ptr [b]

// load b[i]..b[i+4]

movdqa xmm1,  [ebx + eax*4]

pop ebx

movdqa xmm4,  [ecx + eax*8+16]

movdqa xmm5,  [edx + eax*8+16]

movdqa xmm6,  [edx + eax*8]

movdqa xmm7,  [ecx + eax*8]

mov ebx,0

inner:

mov edi, eax

add edi, ebx

push eax

mov eax, dword ptr [a]

// load a[j]..b[j+4]

movdqa xmm0,  [eax + ebx*4]

pop eax

// There is an intricate little dance below to:

// 1)Do the minimal number of PSHUFDs

// 2)Do the minimal number of loads, stores

// 3)Put the maximal distance between obviously

//   dependent instructions.

// 4)Make sure the result array pieces are 16-byte

//   aligned

// We are operating on permutations and combinations

// of a 4x4 section of the multiplication

//

// The technique for minimizing PSHUFD is to

// alternate the use of the target registers

// for the PMULUDQ instruction so that we are

// temporarily through with a register before 

// we step on it.

// 

pshufd xmm3, xmm1, 0x0

pshufd xmm2, xmm0, 0x32

pmuludq xmm3, xmm2

paddq xmm4, xmm3

pshufd xmm3, xmm1, 0x11

pmuludq xmm2, xmm3

paddq xmm5, xmm2

pshufd xmm2, xmm0, 0x10

pmuludq xmm3, xmm2

paddq xmm6, xmm3

movdqa  [edx + edi*8],xmm6

movdqa xmm6,  [edx + edi*8 + 32]

pshufd xmm3, xmm1, 0x0

pmuludq xmm3, xmm2

paddq xmm7, xmm3

movdqa  [ecx + edi*8],xmm7

movdqa xmm7,  [ecx + edi*8 + 32]

pshufd xmm3, xmm1, 0x22

pmuludq xmm3, xmm2

paddq xmm4, xmm3

movdqa  [ecx + edi*8+16],xmm4

movdqa xmm4,  [ecx + edi*8+48]

pshufd xmm3, xmm1, 0x33

pmuludq xmm2, xmm3

paddq xmm5, xmm2

movdqa  [edx + edi*8+16],xmm5

movdqa xmm5,  [edx + edi*8+48]

pshufd xmm2, xmm0, 0x32

pmuludq xmm3, xmm2

paddq xmm6, xmm3

pshufd xmm3, xmm1, 0x22

pmuludq xmm3, xmm2

paddq xmm7, xmm3

add ebx,4

cmp ebx, DIGITS

jl inner

movdqa [edx + edi*8 + 32], xmm6

movdqa [ecx + edi*8 + 32], xmm7

add eax,4

cmp eax, DIGITS

jl outer

// Finally, we do the 29 bit carry calculation

// and add the two result arrays

pxor mm6,mm6

mov eax, 0x1fffffff

movd mm7, eax

movq mm0,  [ecx]

movq mm6, mm0

pand mm0, mm7

movd  [ecx], mm0

psrlq mm6, 29

mov ebx,1

cloop:

movq mm0,  [ecx + ebx*8]

movq mm1,  [edx + ebx*8-8]

paddq mm0, mm1

paddq mm0, mm6

movq mm6, mm0

pand mm6, mm7

movd  [ecx + ebx * 4], mm6

psrlq mm0, 29

movq mm6,  [ecx + ebx*8+8]

movq mm1,  [edx + ebx*8-0]

paddq mm6, mm1

paddq mm6, mm0

movq mm0, mm6

pand mm0, mm7

movd  [ecx + ebx * 4+4], mm0

psrlq mm6, 29

add ebx,2

cmp ebx,DIGITSTIMESTWO-1

jl cloop

sub ebx,1

movd  [ecx + ebx * 4], mm6

}

return 0;

}

付録 A - パフォーマンス・データ

パフォーマンス・データの改訂履歴

表 1:  大数の乗算のパフォーマンス・データ

パフォーマンス・データ(マイクロ秒)
乗算の数/
サイズ

インテル® Pentium® III
プロセッサ

(733 MHz)

インテル® Pentium 4
プロセッサ

(1.20 GHz)
Cの標準的なアルゴリズム 1.16×16 37.5 30.0

フルに最適化された SSE2
(Streaming SIMD Extensions 2)
ASM

2.32×32 なし 2.80

表 2:  表 1のパフォーマンス・データに基づく処理速度の比較

コードおよびプラットフォーム 処理速度

Pentium 4プロセッサ(SSE2 ASMと標準的な Cアルゴリズム) 10.7倍

Pentium 4プロセッサ上の SSE2 ASMと Pentium IIIプロセッサ上の標準的な C 13.4倍

パフォーマンスは、通常のキャッシュ環境を想定して測定された。このアルゴリズムのパ

フォーマンスには、キャッシュは重要な要因ではない。パフォーマンスは、Pentium® IIIプロ
セッサ 733 MHzと Pentium 4プロセッサ 1.20 GHzを使用して測定された。テスト・システムに
ついての詳細は、A-2ページの「テスト・システムの構成」を参照のこと。

この表は、大数の乗算を実行する 2つのアルゴリズムのパフォーマンスの比較をまとめたもの
である。第 1のアルゴリズムは、16ビット乗算を使用する。この結果は、特別な最適化を行わ
ない標準的なアルゴリズムのパフォーマンスを示す。第 2のアルゴリズムは、すべての部分積
が計算されるまでキャリーの伝搬を遅延する手法を使ってフルに最適化されたものである。こ

のアルゴリズムは、本書で説明した 29ビット手法も使用している。

オペランドのサイズを 2倍にするとパフォーマンスが 4倍になり、SIMDを使用して 2つの操
作を並列で実行することでパフォーマンスが 2倍になるとすれば、理論的にはパフォーマンス
が 8倍にアップするはずである。実際には、8倍を超えるパフォーマンスが得られた。これは
次のような理由による。

• 乗算のサイズを 2倍に(16×16から 32×32に)すると、理論的にはパフォーマンスが 4倍に
アップするはずであるが、実際の結果は 4倍を超えている。これは、オペランドが大きい
乗算の方が、オーバーヘッド命令(ロードやストアなど)の割合が小さくなるためである。

• 最適化されたアルゴリズムは、Pentium 4 プロセッサのアーキテクチャに合わせて、きめ細
かくチューニングされている。

テスト・システムの構成

表 3:  Pentium III プロセッサのシステム構成

プロセッサ Pentium IIIプロセッサ 733 MHz
システム インテル® Desktop Board VC820

BIOSのバージョン VC82010A.86A.0028.P10

2次キャッシュ 256 KB

メモリ・サイズ 128 MB RDRAM PC800-45

Ultra ATAストレージ・
ドライバ

製品版候補 6.00.012

ハードディスク IBM DJNA-371800 ATA-66

ビデオ・コントローラ/
バス

Creative Labs 3D Blaster† Annihilator† Pro AGP
nVidia GeForce256† DDR –32MB

ビデオ・ドライバの

リビジョン

NVidiaリファレンス・ドライバ 5.22

オペレーティング・

システム

Windows† 2000ビルド 2195

表 4: Pentium 4 プロセッサのシステム構成

プロセッサ Pentium 4プロセッサ 1.20 GHz
システム インテル Desktop Board D850GB

BIOSのバージョン GB85010A.86A.0014.D.0007201756

2次キャッシュ 256 KB

メモリ・サイズ 128 MB RDRAM PC800-45

Ultra ATAストレージ・
ドライバ

製品版候補 6.00.012

ハードディスク IBM DJNA-371800 ATA-66

ビデオ・コントローラ/
バス

Creative Labs 3D Blaster Annihilator Pro AGP
nVidia GeForce256 DDR –32MB

ビデオ・ドライバの

リビジョン

NVidiaリファレンス・ドライバ 5.22

オペレーティング・

システム

Windows 2000ビルド 2195

